
Abstract

The Action Notation Modeling Language (ANML)
is being developed to provide a high-level, conve-
nient, and succinct alternative to existing planning lan-
guages such as PDDL, theEUROPA modeling lan-
guage (NDDL), and theASPEN modeling language
(AML). ANML is based on strong notions of action
and state (like PDDL and AML), uses a variable/value
model (like NDDL and AML), supports rich temporal
constraints (like NDDL and AML), and provides sim-
ple, convenient idioms for expressing the most common
forms of action conditions, effects, and resource usage.
The language supports both generative and HTN plan-
ning models in a uniform framework and has a clear,
well-defined semantics. In this paper, we highlight the
central and unique features of the ANML language.

1. Introduction

There are a number of different languages that have
been developed and used at NASA for modeling plan-
ning domains and problems. Chief among these are the
NDDL language [1] developed for the EUROPA2 plan-
ner at Ames Research Center, and the ASPENModeling
Language (AML) [2] developed at JPL. While both of
these languages have their strong points, they are quite
different in their basic concepts and assumptions, and
both have significant weaknesses.

The NDDL language facilitates the description of
constraints among intervals. It characterizes every state
or activity as an interval and specifies the possible tem-
poral relationships between those intervals. However,
NDDL has no notion of action, state, fact, or goal. A
planning problem in NDDL simply involves filling out
the set of timelines with intervals so that there are no
gaps and all the constraints are obeyed. NDDL mod-
els are often quite verbose, unintuitive, and contain re-
dundant constraints. It is also easy to make modeling
mistakes in the language, and debugging models is dif-
ficult. In contrast, AML is based on fairly intuitive no-
tions of action and state, and contains many convenient
constructs for describing resource usage and temporal
constraints. However, the language is fundamentally
geared towards HTN planning, and does not fully sup-

port generative planning.
In contrast to these NASA languages, the academic

community continues to develop and use the PDDL
family of languages. PDDL1.2 [3] is based on a
STRIPS model of action, so there is no ability to model
time, duration, concurrency, numeric resources, or tem-
poral constraints. The successors, PDDL2.1 [4] and
PDDL2.2 [5]), introduceddurative actions, and con-
tinuous numeric variables. These languages therefore
permit simple modeling of time, concurrency, and re-
sources. However, the languages are still primarily
proposition based, and do not directly support more
complex temporal conditions, effects, or constraints.
It is also cumbersome to express common patterns of
change and resource usage, and it is easy to make errors
in doing so [6]. PDDL3.0 [7] has introduced the ability
to model more complex temporal constraints and pref-
erences on goals, but the capabilities do not extend to
actions or conditions. Despite these limitations, PDDL
has become the de facto standard in the academic plan-
ning community and is used in the biennial International
Planning Competitions [8, 9, 10]. As a result, many ex-
ample domains are available in PDDL, and many effec-
tive planners and techniques have been developed for
different versions of the language.

We have been developing the Action Notation Mod-
eling Language (ANML) in an effort to 1) provide a
high level, convenient, and succinct alternative to ex-
isting languages, 2) support both generative and HTN
planning models in a uniform framework, 3) provide a
language with clear, well-defined semantics, and 4) al-
low greater compatibility with the evolving PDDL fam-
ily of languages. ANML is based on strong notions
of action and state (like AML and PDDL), uses a vari-
able/value representation (like NDDL and AML), sup-
ports rich temporal constraints (like NDDL and AML),
and provides simple, convenient idioms for expressing
the most common forms of action conditions and ef-
fects. As an example action description in ANML, we
could express a simple high-level navigate action for a
rover as shown in Figure 1. Navigate has two location
parametersfrom andto, and a fixed duration of 5. The
temporal qualifierover all states that the three follow-
ing statements apply to the entire duration of the ac-
tion. The first is a condition stating that the arm must



action Navigate (location from, to) {
duration := 5 ;
over all { arm == stowed ;

position == from ::= to ;
batterycharge :consumes 2.0 } }

Figure 1: A simple ANML action.

(:durative-action navigate
:parameters (?from - location ?to - location)
:duration (= ?duration 5)
:condition (and (at start (position ?from))

(at start (stowed))
(over all (stowed))
(at start (>= (batterycharge) 2.0))
)

:effect (and (at start (decrease (batterycharge) 2.0))
(at start (not (position ?from)))
(at end (position ?to))
))

Figure 2: The equivalent PDDL2.1 action.

remain stowed over the entire action. The second is a
combination of a condition and two effects stating that
the location must initially be the locationfrom, is unde-
fined in the interim, and will be the locationto at the
end of the action. The third is an effect stating that
the action consumes two units of energy. Among other
things, more complex temporal qualifiers are possible
and more complex functional expressions are possible
for duration and energy consumption.

For comparison purposes, Figure 2 shows an equiv-
alent model in PDDL2.1. There are a number of syn-
tactic differences between the two descriptions, but
these are relatively unimportant. The more signifi-
cant differences in both size and simplicity are due
partly to ANML’s variable/value representation, but
also to ANML’s more powerful constructs for describ-
ing change. In the ANML description, we have not par-
titioned the statements into conditions and effects. In-
stead, we have described what happens to each relevant
variable in turn.

In the sections that follow, we explain these capabil-
ities in greater detail, and highlight the novel features
of the ANML language. We focus on the powerful and
concise constructs in ANML for temporal qualification,
for describing change over the course of an action, for
describing resource usage, and for integrating task de-
composition with traditional action models. To do this,
we must first introduce the basic entities of the ANML
language, time varyingvariablesandfunctions.

2. Basic Declarations

Constants, variables, and functions in ANML are
typed. There are a number of built-in types, includ-
ing: int, float, bool, string, symbol, object, vector, vari-
able, function, andaction. There is no need to declare
numeric or string constants in ANML, but objects and

symbols must be declared. For example:

object spirit, opportunity, rock1, sample2 ;

declaresspirit, opportunity, rock1, and sample2 to be
constants of type object.1 It is also possible to declare
new types in ANML. For example:

type positiveInt float [0.0, inff] ;
type rover object {spirit, opportunity, pathfinder} ;
type color symbol {blue, green, red, yellow, purple } ;

defines the typespositiveInt, rover, andcolor as special-
izations of the typesInt, object, andsymbol respectively.
It is also possible to define specialized vector types, e.g:

type location vector( positiveInt x, y ) ;
type path vector( location from, to ) ;

which defines locations to be vectors of two integer ele-
ments,x andy, and paths to be vectors of two locations,
from andto. ANML also allows the definition of more
complex structured types, but we will not discuss this
further here.

2.1 Variables

When declaring a variable in ANML, one must spec-
ify the domain of the variable. The declaration there-
fore consists of the keywordvariable followed by a type,
followed by the variable name. Any predefined or user-
defined type can be used for this purpose. For example:

variable bool havePicture ;
variable string sampleName, pictureName ;
variable positiveInt [0, 5] sample# ;
variable float [0.0, inff] batterycharge, wheelCurrent ;
variable color {blue, red, green} filterColor ;
variable rover r ;
variable location position ;

are all legitimate variable declarations.
For convenience, variables can be initialized in a dec-

laration. For example:

variable float roverSpeed := 30.0 ;

is equivalent to stating:

variable float roverSpeed ;
at start { roverSpeed := 30 } ;

Vectors can also be initialized in this way. For example:

variable vector( int x, y ) position := (5, 20) ;

is equivalent to stating:

variable vector( int x, y ) position ;
at start { position.x := 5 ;

position.y := 20
}

1As far as ANML is concerned,object andsymbol are
synonymous, but the former is typically used for tangible en-
tities like spacecraft and rovers, while symbol is typically used
for intangibles like colors or shapes.



2.2 Functions

Functions in ANML are essentially parameterized
variables. So when declaring a function symbol, in
addition to specifying the range of the function, one
must also specify the domains of the parameters or argu-
ments. The declaration for a function symbol therefore
consists of the keywordfunction followed by a type, fol-
lowed by a function symbol and its typed argument list.
For example:

function float [0, inff] batterycharge(rover r) ;

indicates that the functionbatterycharge of a rover is a
positive float. As with variables, functions in ANML
are implicit functions of time. In fact, variables in
ANML can be considered as functions having zero ar-
guments. Unlike variables, the value of a function can-
not be assigned in the declaration, because doing so
might require different assignments for different argu-
ments.

Predicates in ANML are simply functions onto the
boolean valuestrue andfalse. Thus, the statements:

predicate havesample ;
predicate stowed(instrument i) ;

are equivalent to stating:
variable bool havesample ;
function bool stowed(instrument i) ;

3. Actions

Actions are the means by which one changes the
world. In general, an ANML action description consists
of:

• the action name
• a typed parameter list (optional)
• a duration assignment (optional)
• local variable or function declarations (optional)
• one or more temporally qualified conditions, ef-

fects, or change statements

In Figure 1 we showed an example of a simple navigate
action in ANML:

action Navigate (location from, to) {
duration := 5
over all { arm == stowed ;

position == from ::= to ;
batterycharge :consumes 2.0 } }

The action name and typed parameter list are shown in
the first line, and the duration assignment is shown in
the second line. The action does not contain any local
variable definitions, but we could have done something
like define a local variable for energy use in terms of
duration, and then express the consumption using this
local variable:

variable float energy := duration * use-rate ;
over all { batterycharge :consumes energy } ;

The main body of our example action consists of the
three lines:

over all { arm == stowed ;
position == from ::= to ;
batterycharge :consumes 2.0 } ;

In general, the body can express simple conditions (like
the first line above), simple effects, or more complex
combinations of conditions and effects (lines 2 and 3).
To be more precise, these statements take the form:

Temporal Qualifier{φ1; . . . ;φn}
whereTemporal Qualifieris something likeat start or
over all, and each of theφi is a condition, effect, or
change expression. A simple condition expression is
of the form:

variable relation expression
whererelation is typically==, in, or a numeric compara-
tor (<, ≤, >, ≥). The expression is frequently a con-
stant, or another variable, but can be a set, interval or
an algebraic expression of numeric variables and con-
stants.

A simple effect expression is of the form:
variable assignment expression

whereassignmentis one of:= or :in. The expression is
the same as for conditions, but can also beundefined
(it is not allowed to condition on a variable beingun-
defined). Conditions and effects can be distinguished
because the relations in conditions are distinct from the
assignment operators used in effects. Because of this,
there is no need to separate them or delineate them with
a keyword as in PDDL.

Combinations of conditions and effects only make
sense over intervals. In general, they are of the form:

variable relation expression1
assignment expression2
:assignment expression3 ;

The relation and expression in the first line expresses a
condition that must hold at the beginning of the interval.
For example, in the expression:

over all { position == from
:= undefined
::= to } ;

the condition isposition == from, which must hold at
the beginning of the interval. This condition expression
may be omitted, which would indicate that there is no
requirement on the initial value of the variable. The
assignment and expression in the second line is an ef-
fect indicating the value of the variable over the interior
of the interval. This assignment can also be omitted,
in which case, the variable is assumed to be undefined
during the interval. The last assignment and expression
is also an effect indicating the final value of the vari-
able at the end of the interval. In the example above,
the final effect::= to specifies that the position will be
the destinationto at the end of the interval. Note that
the assignment operator for a final effect is preceded by
an additional colon to distinguish it from an interim ef-
fect. Thus the above expression could be expressed as a
simple condition and two simple effects:



at start { position == from } ;
over interim { position := undefined } ;
at end { position := to } ;

3.1 Temporal Qualifiers

A temporal qualification indicates the time or time
period over which a variable has a particular value or
changes its value. In the example above, we saw the
use of the temporal qualifiers:over all, at start andat
end. These qualifiers are special cases of a more general
temporal qualifierin, and can also refer to time points or
intervals other than just the start, end, or entire duration
of an action. The general form for the temporal qualifier
in is:

in i dur d Φ ;

wherei is an interval of time,d is a duration andΦ is
a collection of conditions. It means thatΦ must hold
for at least the durationd within the intervali. If the
duration is omitted,Φ need only hold for some instant
within the specified interval. As an example, the condi-
tion:

in [start+5,end-2) dur 3 { heater == on } ;

specifies the condition that the heater must be on for at
least 3 time units between the start of the action plus 5
time units and the end of the action minus 2 time units.
The time interval can be closed or open at either end,
and time points are specified relative tostart andend,
meaning the start and end of the action respectively. The
keyword all is short for the closed interval[start,end]
and the keywordinterim is short for the open interval
(start,end). The full power of the qualifierin is allowed
for specifying conditions. However, if used for effects
or change it could allow the expression of uncertainty.
We therefore limit its use to pure conditions. For effects
and change, we are limited to the two special casesat
andover. The temporal qualifierat is defined as:

at t Φ ≡ in [t,t] Φ ;

and means thatΦ must hold or take place at the time
instantt. As with in, t can be any time point relative to
the start or end of the action.

Similarly, the temporal qualifierover is defined as:

over i Φ ≡ in i dur ‖i‖ Φ ;

where‖i‖ refers to the length of the intervali. As with
in, the intervali can be closed or open at either end, and
the time points are specified relative to the start or end
of the action.

3.2 Relative Change

For numeric variables, it is often useful to specify ef-
fects relative to the existing value, rather than in abso-
lute terms. For example we might want to specify that a
particular drilling operation advances the drill a certain
amount beyond its current depth. We could state this as:

at end { depth := depth + increment } ;

Some languages make this a bit easier by allowing addi-
tional operators like+= and–=. In ANML we do this by
specifying change on the “delta” of the depth variable
rather than on the depth variable itself. For the above
example we would say:

at end { 4depth := increment } ;

The meaning of this is that thedepth variable changes
by the amountincrement.2 The reason we take this
approach is that it allows us to conveniently say other
more difficult things like:

over all { 4depth ::= increment } ;
over all { 4depth :in [0, increment] ::= increment } ;

The first of these implies that the change in depth is un-
defined over the course of the interval, before taking on
the final valueincrement. The second implies that the
change in depth is bounded by the interval[0, increment]
over the course of the interval, before taking on the fi-
nal valueincrement. These statements are considerably
more cumbersome using only the basic primitives, be-
cause they require defining a temporary variable to hold
the initial value of the depth variable, e.g:

variable float start-depth := depth ;
over interim {depth := undefined } ;
at end {depth := start-depth + increment } ;

As we will see in the next section, composite incremen-
tal change statements will prove extremely convenient
for describing resource consumption and production.

4. Resources

Resourcesare common and convenient abstractions
in many planning and scheduling applications. There
are many different kinds of resources – they can be
discrete or continuous, and consumable or reusable.3

Discrete resources can also beunit-capacity, or multi-
capacity. The different types of resources are illustrated
in Table 1.

Discrete Continuous
Reusable instruments, bandwidth,

tools power
Consumable solid rockets energy, fuel,

cryogen

Table 1: Resource types.

As far as ANML is concerned, resources are just nu-
meric variables. They are therefore declared in the same
way:

2As a practical matter, the delta symbol (4) is a bit hard
to type on most keyboards, so we use the carat symbol (∧) as
a substitute, e.g.∧depth instead of4depth.

3The first of these dimensions,discrete or continuous, is a
property of the resource variable itself. The second,consum-
able or reusableis a property of how the resource is used. In
fact, it is entirely possible for a quantity to be a reusable re-
source for one action and a consumable resource for another.



variable float [10.0, 100.0] batterycharge := 50.0 ;
variable integer [0, 12] sample-bags := 12 ;

It is the usage of these resources where ANML pro-
vides additional facilities beyond the notation described
so far. We start withreusableresources.

4.1 Usage

A reusableresource is one that isconsumedat the
beginning of an action, but given back (orproduced) at
the end. Using the mechanisms for relative change that
we introduced above, we could express this as:

at start { 4resource := –quantity } ;
at end { 4resource := quantity } ;

However, because resource use is so common, we have
introduced a more convenient way of expressing this
pair of effects:

over all { resource :uses quantity } ;

It is important to note that for a resource effect like this,
the condition:

at start { resource >= quantity } ;

is implicit because of the definition of the resource vari-
able. For example, a declaration ofbatterycharge as:

variable float [10.0, 100.0] batterycharge ;

implicitly requires that the quantity remain within the
interval [10.0, 100.0]. Any action that would violate
these bounds would not be legal.

The counterpart tousinga resource for a period of
time is to make a resource available for a period of time.
For example, by running a generator, we could make ad-
ditional power available. We could simply model this as
negative resource usage, but for convenience and clarity
we refer to this aslendinga resource and express it as:

over all { resource :lends quantity } ;

4.2 Consumption and Production

As we hinted above, resource usage can be thought of
as a pair of consumption and production effects. Using
the mechanisms for modeling relative change, we can
model instantaneous resource consumption as:

at t { 4resource := –quantity } ;

For convenience and clarity we allow this to also be
stated as:

at t { resource :consumes quantity } ;

Similarly, resource production can be modeled as:

at t { 4resource := quantity } ;

or for convenience as:

at t { resource :produces quantity } ;

When dealing with consumption and production of
resources, it is common to make a conservative mod-
eling assumption that consumption occurs at the begin-
ning of an action and production occurs at the end. In

reality, consumption and production usually occur grad-
ually over the course of actions, although the actual
function may be complex or even unknown. The con-
servative discretization ensures that there is enough of a
resource at the beginning of a consumption action, and
that we do not rely on any production actions until their
end. In effect, this approach tracks a lower bound for
the resource, and guarantees that plans will never vi-
olate the lower bound limit for the resource variable.
While this works well in many situations, it can run
into trouble if the resource variable also has an upper
bound, orcapacity, such as for a battery, fuel tank, or
storage container [6]. Figure 3 illustrates the problem.
A navigate action is performed while the battery is be-

final charge

Navigate Action

Recharge Action

Recharge production

Navigate consumption

combined

lower limit

lower bound

capacity

battery charge

Figure 3: Illustration of simultaneous consump-
tion and production activities. Although the lower
bound remains within the allowed range for the re-
source, the actual value exceeds resource capac-
ity unless the consumption activity is started ear-
lier.

ing recharged (by solar panels). The conservative lower
bound envelope for the two actions remains within the
allowed range for the battery. However, since produc-
tion actually occurs before the end of the recharge ac-
tion, the actual charge envelope will exceed battery ca-
pacity. This kind of problem can occur any time there
is the possibility of simultaneous consumption and pro-
duction, and the resource has a capacity limit. Cushing
and Smith [6] discuss this problem in detail, and dis-
cuss some alternative approaches to dealing with it. The
most obvious way of dealing with this problem (without
resorting to detailed modeling and reasoning about con-
tinuous consumption and production) is to keep track of
both a lower bound and an upper bound for a resource.
For example, we could model the behavior of a con-
sumption action as:

at start { 4resource lb := –quantity } ;
at end { 4resource ub := –quantity } ;

and a production action as:

at start { 4resource ub := quantity } ;
at end { 4resource lb := quantity } ;

While sound, this requires that we define and keep track
of two explicit variables for each resource, and get the
timing and polarity of the conditions and effects right.



We regard this as both complex and cumbersome, and
likely to result in many modeling errors. Instead, we
think that the user should be able to simply state that
consumption or production occurs over the course of
an action (or interval). The planner should manage the
details of keeping track of lower and upper bounds for
variables in order to assure soundness. Using the no-
tation for relative change, and for compound change
statements, we would express a typical resource con-
sumption as:

over all { 4resource :in [–quantity,0] ::= –quantity } ;

This states that over the course of the action, the re-
source change is guaranteed to be between zero (upper
bound) and–quantity (lower bound) and will be at the
lower bound at the end of the interval. We define the
shorthand:

over all { resource :consumes quantity } ;

to mean precisely this for an interval; the quantity
change is bounded over the course of the interval, and
ends with the specified change. Similarly, resource pro-
duction over an interval:

over all { resource :produces quantity } ;

is defined as:
over all { 4resource :in [0,quantity] ::= quantity } ;

The:consumes and:produces statements are extraor-
dinarily convenient and powerful, because they allow
the user to specify complex production and consump-
tion activities without requiring details of the actual
consumption and production functions, and without re-
quiring that the user explicitly provide upper and lower
bound discrete approximations. Finally, it is worth not-
ing that for some lengthy activities, like solar recharg-
ing, and driving, it may be necessary to break produc-
tion and consumption into phases so that actions can be
performed concurrently in order to make effective use
of resources, and keep them within bounds. For exam-
ple we could model solar production over the course of
an extended period as a series of production activities:

over [start, start+5) { energy :produces 2.0 } ;
over [start+5, start+10) { energy :produces 3.5 } ;
. . .
over [start+95, end) { energy :produces 1.3 } ;

This allows consumption activities like navigation or
shunting to be performed throughout the production
without violating either lower bound or capacity limits.

5. HTN Decomposition

Hierarchical Task Networks (HTNs) and task decom-
position planning techniques are used for many practi-
cal planning systems and applications [11, 12, 13, 14].
For HTN systems, goals are stated in terms of high level
tasksto be performed, andmethodsallow for the de-
composition of these tasks into lower level tasks and
primitive actions. Many within the planning commu-
nity have argued against the HTN representation and

task decomposition on the grounds that they describe
what actionsshouldbe used for (their purpose), rather
than what actionsdo (their conditions and effects). As
such, this representation can lead to systems that are
highly tailored to a certain class of problems, but can be
brittle if they are asked to solve problems that require
using actions in unanticipated ways. All of this is true.
Nevertheless, there are some good arguments for allow-
ing task decomposition in a modeling language:

1. Some planning domains seem to be more naturally
expressed in terms of task decomposition.

2. There are situations where the modeler may be able
to characterize ways of achieving tasks without hav-
ing a clear understanding of the effects and condi-
tions of the underlying actions.

In developing ANML, we did not wish to unduly con-
strain the modeler, or force the modeler to describe ac-
tions at an unnecessarily fine level of detail. As a re-
sult, ANML allows the specification and use of action
decomposition by allowing decomposition expressions
within action descriptions. We have done this in a novel
way that allows action decomposition to be fully inte-
grated with ordinary action descriptions containing con-
ditions and effects. There are three things necessary to
make this work:

1. Each action (instance)A is regarded as having an im-
plicit effect proposition of the same name (A) over its
execution. In other words:

over all { A == false := true ::= false } ;

2. We allow these “action” propositions to be used as
ordinary conditions within action descriptions.

3. We allow relative ordering constraints on conditions.

Allowing relative ordering constraints among condi-
tions requires some additional machinery which we
now describe.

5.1 Relative Ordering Constraints

Suppose that we have two action conditionsp andq,
that must be true at some point during the action, e.g:

in all { p ; q } ;

Now imagine that we also require thatp become true
before q. Using the machinery presented so far, we
could specify explicit intervals forp and q that guar-
anteed this, but we could not simply specify a relative
ordering constraint between the two conditions. To do
this, we need to be able to name conditions, and refer
to their start and end times. We name conditions by fol-
lowing them with the symbol! and a name, e.g:

in all { p ! c1 ; q ! c2 } ;

To refer to the start and end times of conditions we gen-
eralize the keywordsstart andend to allow them to refer
to the start and end of specific actions, e.g:start(c1). We
can then specify the desired relative ordering constraint
as:



start(c1) < start(c2) ;

The entire action condition would therefore be:

in all { p ! c1 ; q ! c2} ;
start(c1) < start(c2) ;

Alternatively, we could specify the two conditions
separately as:

in all { p ! c1 } ;
in (start(c1), end] { q } ;

This avoids the need to actually name the second condi-
tion q. This ordering capability is quite general and al-
lows us to specify complex temporal constraints among
arbitrary sets of conditions. For convenience, we intro-
duce the shorthandordered(p1, . . . , pk) to refer to an
ordered set of conditions. In other words:

in all { ordered(p1, . . . , pk) } ;

is equivalent to naming each of the conditions, and
specifying temporal constraints between each succes-
sive pair:

in all { p1 ! c1 ; . . . ; pk ! ck } ;
end(c1) ≤ start(c2) ;
. . .
end(ck−1) ≤ start(ck) ;

A similar shorthandunordered(p1, . . . , pk) can be used
to refer to an unordered set of conditions. Together,or-
dered andunordered can be used to describe a partial
ordering of conditions. For example:

ordered( p, unordered( q, ordered( r, s )), t ) ;

corresponds to the partial order shown in Figure 4.

p

q

r s

t

Figure 4: A partially ordered set of conditions.

5.2 Decompositions

With the ability to express relative ordering con-
straints on conditions, we now have the tools necessary
to express decomposition in an action description. As
an example, suppose we want to specify a method for
the high level task ofCollectSample, which can be de-
composed into four primitive actions of unstowing the
instrument, placing the instrument, taking the sample,
and stowing the instrument. We can express this in
ANML as:

action CollectSample(location l) {
over all { position == l } ;
in all { ordered( Unstow, Place(l),

TakeSample(l), Stow ) } ;
. . . } ;

In effect, this says that in order to successfully perform
theCollectSample action, we must successfully perform
theUnstow, Place( ), TakeSample( ), andStow actions,
in order.

6. Related Work

The ANML language has taken inspiration from sev-
eral existing languages. Among these, the NDDL lan-
guage has the most powerful capability for expressing
temporal constraints. However, these capabilities are
not necessarily easy to understand or use, and we find
many aspects of the language to be both cumbersome
and difficult to use effectively. The AML language has
very convenient and natural syntax for expressing re-
source usage, but the HTN nature of the language is lim-
iting. The PDDL family of languages has been carefully
developed, and is widely adopted in the research com-
munity. However, the propositional nature of the lan-
guage, the limited constructs for describing change and
resource usage, and the limited ability to model time
and temporal constraints make it difficult or impossible
to use for serious applications. The change notation in
ANML most closely resembles constructs developed in
the SAS family of languages [15]. However, in overall
capability and style, the IxTeT language [16] is perhaps
the closest.

There have been a few previous attempts to merge
HTN decomposition into more traditional action lan-
guages like PDDL [3, 14]. However, these attempts
have not been widely used or adopted because the two
paradigms have had separate semantics and have not re-
ally been integrated. The semantics we ascribe to de-
compositions in ANML is quite different. Essentially,
we regard a decomposition as simply being another set
of conditions necessary for performing the action. In
other words, if the subtasks can be performed in the or-
der indicated, then the high-level task can be performed.
This makes sense for several reasons:

1. If the decompositions for an action prove to be im-
possible (cannot be performed), then the action itself
is not possible (or we do not know its outcome).

2. Multiple decompositions correspond to a disjunction
of sets of conditions, and any one of these sets would
be sufficient to accomplish the action.

3. Logical conditions can be mixed with decomposi-
tions.

4. It is consistent with, and can be seen as a generaliza-
tion of allowing general temporal constraints among
conditions.

7. Conclusions and Future Work

In this short paper, we have only sketched some of
the key features of the ANML language. In particu-
lar, we have described the powerful and concise con-
structs in ANML for temporal qualification, for describ-



ing change over the course of an action, for describing
resource usage, and for integrating task decomposition
with traditional action conditions and effects. There are
additional features and details of the language that we
have not mentioned, or have only glossed over. Among
other things, there is the ability to:

• define structured objects
• express quantification
• express disjunctive conditions
• express conditional effects
• express complex goals

A draft manual describing the language is available
by request, and we are in the process of building trans-
lators from PDDL into ANML and from ANML into
NDDL. Because of the expressiveness of ANML, only
a subset of the language can be translated into AML,
but this is also being considered.

There is an additional effort underway to extend the
syntax of ANML to allow description of continuous
change, processes and exogenous events. We would like
to be able to express resource usage as a function of time
where known and convenient. However, our view here
is not that a planner must necessarily be able to reason
about continuous change. Instead, our view is that we
should allow the user to naturally express the domain,
at whatever level of detail is appropriate. It is then per-
fectly reasonable for a planner to make sound but in-
complete approximations to effectively reason about the
domain. As a case in point, reasoning about lower and
upper bound envelopes is a useful approximation that is
both computationally tractable, and perfectly adequate
for many domains. However, we want the planner to
be able to choose the approximation, rather than forc-
ing the user to encode it explicitly. Thus in the case
of continuous change, we would like to see the planner
choose an appropriate discretization or linearization in
order to enable sound, but effective, reasoning about the
domain.

